
1) Overview

This document lists all the changes needed to correct problems in the TTCN implementation of test case 6.3.3, which is part of the LTE test suite. 

The test case can be demonstrated to run with one LTE UE (see section 6). Execution log files are provided as evidence. 

3) Verification Test Summary

Test Case:
TC_6_3_3
Test Group:
LTE_IRAT\6_3
ATS Version:
iwd-EUTRA-B2013-03_D13k43
System Simulator used:
Anritsu Protocol Conformance Test System ME7832L
UE used:
Nvidia Icera i500
Verification Status:
PASS
4) Corrections required for test case 6.3.3
Introduction

This section describes the changes required to make test 6.3.3 run correctly with a LTE UE. The TTCN used to pass this test case belongs to iwd-EUTRA-B2013-03_D13k43 release.

Change 1

	Testcase name
	6.3.3

	Reason for change
	According to test case specification 36.523-1 b40:
-
System information combination 4 as defined in TS 36.508 [18] clause 4.4.3.1 is used in E-UTRA Cell 1.

-
System information combination 11 as defined in TS 36.508 [18] clause 4.4.3.1 is used in E-UTRA Cell 2.

For change sysinfo combination is used function f_EUTRA_ChangeSysinfoCombination. Current implementation of this function set SIB6 to omit which causing issue later in function f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8 and TTCN fail with error message "FDD template for frequency f8 cannot be set". This occurs only on cell1 where sysinfo combination was change from c11 to c4.

According to test case specification 36.523-1 b40 Table 6.3.3.3.3-5:
SIB6 should set with same values as in cell2 with sysinfo combination c11.

	Summary of change
	· Add new variable v_sib6.
· Get sib 6 before change sysinfo combination – cell1, combination c4

· Set sib 6 after  change sysinfo combination

	Source of change
	LTE_IRAT\6_3\CSG_TestcasesUG.ttcn


Before change:

	  function f_TC_6_3_3_EUTRA() runs on EUTRA_PTC

  {

    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;

    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;

    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;

    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;

    var template (value) CellPowerList_Type v_CellPowerList_Postamble;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var GutiParameters_Type v_GutiParams;

    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;

    var CellReselectionPriority v_CellReselectionPriority_Cell5 := 5;

    var boolean v_CSInOtherRAT := false;

    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;

    timer t_WaitTime := tsc_IdleMode_GenericTimer;

    v_CellPowerList_T0toT1 := {

      cs_CellPower(eutra_Cell1, -91),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T1toT2 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T2toT3 := {

      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T3toT4 := {

      cs_CellPower(eutra_Cell1, -91),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_Postamble := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE)

    };

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init(c11);

    //Set system information combination 4 in cell 1

    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c4);


After change :

	  function f_TC_6_3_3_EUTRA() runs on EUTRA_PTC

  {

    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;

    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;

    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;

    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;

    var template (value) CellPowerList_Type v_CellPowerList_Postamble;

    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;

    var GutiParameters_Type v_GutiParams;

    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;

    var CellReselectionPriority v_CellReselectionPriority_Cell5 := 5;

    var boolean v_CSInOtherRAT := false;

    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;

    timer t_WaitTime := tsc_IdleMode_GenericTimer;

    var template (value) SystemInformationBlockType6 v_Sib6; 

    v_CellPowerList_T0toT1 := {

      cs_CellPower(eutra_Cell1, -91),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T1toT2 := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),

      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)

    };

    v_CellPowerList_T2toT3 := {

      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_T3toT4 := {

      cs_CellPower(eutra_Cell1, -91),

      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)

    };

    v_CellPowerList_Postamble := {

      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE)

    };

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init(c11);

    //Set system information combination 4 in cell 1

    v_Sib6 := f_EUTRA_CellInfo_GetSib6(eutra_Cell1); 

    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c4);

    f_EUTRA_CellInfo_SetSIB6(eutra_Cell1, v_Sib6);     


Change 2

	Function name
	f_EUTRA_CellInfo_GetSib6

	Reason for change
	Function f_EUTRA_CellInfo_GetSib6 handles only case when sysinfo combination is set to c4 or c10. 
SIB6 is in different place in cell info structure when other combination c9, c10a, c11 is used. 
Function needs to be updated for sysinfo combination c9, c10a, c11.

	Summary of change
	Rewrite function f_EUTRA_CellInfo_GetSib6 and add cases when sysinfo combination c9, c10a, c11 are used.

	Source of change
	Common\EUTRA_CellInfo.ttcn


Before change:

	  function f_EUTRA_CellInfo_GetSib6(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return SystemInformationBlockType6
  {

    var EUTRA_CellSysInfo_Type v_SysInfo := f_EUTRA_CellInfo_GetSYSINFO(p_CellId);

    return v_SysInfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib6;
  }


After change :

	  function f_EUTRA_CellInfo_GetSib6(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC return template (value) SystemInformationBlockType6  

  {

    var EUTRA_CellSysInfo_Type v_SysInfo := f_EUTRA_CellInfo_GetSYSINFO(p_CellId);

    var template (value) SystemInformationBlockType6 v_SIB6;
    select (v_SysInfo.Combination) {

      case (c4, c10) {                              

        v_SIB6 := v_SysInfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib6;

      }

      case (c9, c10a) {

        v_SIB6 := v_SysInfo.BCCH_Info.SIs[2].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[1].sib6;        

      }

      case (c11) { // @sic R5-103363 sic@

        v_SIB6 := v_SysInfo.BCCH_Info.SIs[3].message_.c1.systemInformation.criticalExtensions.systemInformation_r8.sib_TypeAndInfo[0].sib6;

      }

    }    

    return v_SIB6;


Change 3

	Testcase name
	6.3.3

	Reason for change
	1, At the begining of TC TTCN modifies SIB1 on cell2. During SIB1 modification, cell2 is set to off and UE is switch off too. According to specification 36.523-3 chap. 7.4.5:

Cells being switched off need not to be mapped to physical resources.
Is not needed to use function f_EUTRA_ModifySysinfoUE_Off and calculate modification interval because cell2 is set to off. Function f_EUTRA_SS_ConfigureSysinfo could by used by TTCN to modify SIB1. 
2, In current TTCN implementation, EUTRA removes CSG list in USIM and in same time UTRAN expects attach procedure. It is caused by wrong coordination triggers. UTRAN side should wait for CSG list to be emptied in LTE side.
3, Fifth input parameter (p_PreambleOnLTE) to function in step 40-46 should be set to true. UE has performed Attach procedure to cell2 in step16.  

	Summary of change
	1, Use function f_EUTRA_SS_ConfigureSysinfo  instead of function f_EUTRA_ModifySysinfoUE_Off 
2, Instead of f_IRAT_WaitForCoOrd_Trigger should be used f_IRAT_SendCoOrd
3, instead of false use true – fifth input parameter step 40-46

	Source of change
	LTE_IRAT\6_3\CSG_TestcasesUG.ttcn


Before change:

	    // Empty UE CSG list in USIM

    f_EUTRA_Preamble_EmptyCSG_AllowedList(eutra_Cell2);

    // Broadcast correct CSG id

    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);

    f_EUTRA_ModifySysinfoUE_Off(eutra_Cell2); //NOTE1
    // Coordination to start preamble in UTRAN

    f_IRAT_WaitForCoOrd_Trigger(UTRAN); //NOTE2
    f_EUTRA_TestBody_Set (true);

    // Since Authentication parameters were updated in the idle updated procedure on UTRAN side...

    //...receive the updated parameters

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    UTRAN.receive (cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG;

    //@siclog "Step 14" siclog@

    //Change Cell 1 and Cell 2 power levels according to the row "T1" in table 6.3.3.3.2-1.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);

    //@siclog "Step 15" siclog@

    //Perform manual CSG selection of cell 2

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);

    f_UT_ManualPLMN_CSG(UT, f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity), v_CSG_Identity_Cell2);

    //@siclog "Step 16" siclog@

    //Check that UE camps on Cell 2 with the generic test procedure TS 36.508 table 6.4.2.7A-2

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT(eutra_Cell2, v_CSInOtherRAT, false, false, v_IRAT_Coordination_MSG.OctetData);

    //@siclog "Step 16A" siclog@

    //Wait 6 seconds for UE to receive system information

    f_Delay (6.0);

    //@siclog "Step 17" siclog@

    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T2" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_EUTRA_SendAuthParameters(eutra_Cell2, UTRAN);

    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);

    f_IRAT_WaitForCoOrd_Trigger (UTRAN);

    //@siclog "Step 28" siclog@

    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    //@siclog "Step 39" siclog@

    //Change Cell 1 and Cell 2 level according to row "T4" in table 6.3.3.3.2-1.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);

    t_WaitTime.start;

    //@siclog "Step 40-46" siclog@

    //Check that UE camps on Cell 1 with the generic test procedure TS 36.508 subclause 6.4.2.7A

    f_EUTRA_TrackingAreaUpdateFromAnotherRAT(eutra_Cell2, tsc_RRC_TI_Def, v_CSInOtherRAT, false, false, v_IRAT_Coordination_MSG.OctetData, false, NORMAL); //NOTE3
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.3.3 Step 46");

    t_WaitTime.stop; // If this times out it will be picked up in the default

    f_EUTRA_TestBody_Set (false);


After change :

	    // Empty UE CSG list in USIM

    f_EUTRA_Preamble_EmptyCSG_AllowedList(eutra_Cell2);

    // Broadcast correct CSG id

    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);

    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2); //NOTE1
    // Coordination to start preamble in UTRAN

    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger); //NOTE2
    f_EUTRA_TestBody_Set (true);

    // Since Authentication parameters were updated in the idle updated procedure on UTRAN side...

    //...receive the updated parameters

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    UTRAN.receive (cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG;

    //@siclog "Step 14" siclog@

    //Change Cell 1 and Cell 2 power levels according to the row "T1" in table 6.3.3.3.2-1.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);

    //@siclog "Step 15" siclog@

    //Perform manual CSG selection of cell 2

    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);

    f_UT_ManualPLMN_CSG(UT, f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity), v_CSG_Identity_Cell2);

    //@siclog "Step 16" siclog@

    //Check that UE camps on Cell 2 with the generic test procedure TS 36.508 table 6.4.2.7A-2

    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT(eutra_Cell2, v_CSInOtherRAT, false, false, v_IRAT_Coordination_MSG.OctetData);

    //@siclog "Step 16A" siclog@

    //Wait 6 seconds for UE to receive system information

    f_Delay (6.0);

    //@siclog "Step 17" siclog@

    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T2" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_EUTRA_SendAuthParameters(eutra_Cell2, UTRAN);

    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);

    f_IRAT_WaitForCoOrd_Trigger (UTRAN);

    //@siclog "Step 28" siclog@

    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);

    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);

    //@siclog "Step 39" siclog@

    //Change Cell 1 and Cell 2 level according to row "T4" in table 6.3.3.3.2-1.

    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);

    t_WaitTime.start;

    //@siclog "Step 40-46" siclog@

    //Check that UE camps on Cell 1 with the generic test procedure TS 36.508 subclause 6.4.2.7A

    f_EUTRA_TrackingAreaUpdateFromAnotherRAT(eutra_Cell2, tsc_RRC_TI_Def, v_CSInOtherRAT, false, true, v_IRAT_Coordination_MSG.OctetData, false, NORMAL);  //NOTE3
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.3.3 Step 46");

    t_WaitTime.stop; // If this times out it will be picked up in the default

    f_EUTRA_TestBody_Set (false);


Change 4

	Testcase name
	6.3.3

	Reason for change
	1, According to test case specification 36.523-1 b40:
Cell 5 and Cell 7 have different Routing Area.
New RAC is set for Cell7 cell info but is not set to SIB1. Is needed to update SIB1 because it caused problem during RAU procedure step18-27.
2, In current TTCN implementation, EUTRA component removes CSG list in USIM and UTRAN component expects attach procedure in same time. It is caused by wrong coordination triggers. UTRAN side should wait for CSG list to be emptied in LTE side.
3, According to test specification UE is registered to cell5. TTCN expects RAU procedure to cell7 but cell7 is not prepared for RRC connection establishment – step 18-27. TTCN expects second RAU procedure to cell5 but cell5 is not prepared for RRC connection establishment too – step 29-38.

	Summary of change
	1, Add new variable v_SIB1_Cell7, read current SIB1 from cell7, updates SIB1 with new RAC and set updated SIB1.
2, Instead of f_IRAT_SendCoOrd should be used f_IRAT_WaitForCoOrd_Trigger
3, Add function f_UTRAN_PrepareCellRRC_ConnEst to step 18-27 and step 29-38. Need this step as MAC dedicated was connected to cell 5/7 and now needs to move to cell7/5.

	Source of change
	LTE_IRAT\6_3\CSG_TestcasesUG_UTRAN.ttcn


Before change:

	function f_TC_6_3_3_UTRAN ( ) runs on UTRAN_PTC {

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());

    timer t_WaitTime := tsc_IdleMode_GenericTimer_UTRAN;

    //Initialise all cells

    f_UTRAN_Init (EUTRA_UTRAN);

    //Set RAC of cell 7 to different from RAC of cell 5 according to Pre-test condition: "Cell 5 and Cell 7 have different Routing Area."

    f_UTRAN_CellInfo_SetRAC(utran_Cell7, tsc_RAC_2);

    //Send the cell info to the EUTRA component

    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);

    //Get Cell Parameters for EUTRA cell

     v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);

    //Update SIB19 of cell 5 according to Table 6.2.3.3.3.3-1

    f_UTRAN_SysInfo_SetSIB19 (utran_Cell5,

                              cs_SIB19_UtraAndOneEutraEntry (cs_Utra_PriorityInfoList_Def (v_UtraPriorityValue),

                                                             cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                   v_EUTRA_MeasurementBandwidth)));

    f_UTRAN_SysInfo_SetSIB19 (utran_Cell7,

                              cs_SIB19_UtraAndOneEutraEntry (cs_Utra_PriorityInfoList_Def ( v_UtraPriorityValue ),

                                                             cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                   v_EUTRA_MeasurementBandwidth)));

    //Create and configure Cell 5 & 7

    //Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellFACH (utran_Cell5);

    f_UTRAN_SS_CreateCellFACH (utran_Cell7);

    //Start sending System Information

    f_UTRAN_SendDefSysInfo (utran_Cell5);

    f_UTRAN_SendDefSysInfo (utran_Cell7);

    // Wait for CSG list to be emptied in LTE side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger); //NOTE2
    //Set power levels according to T0

    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);

    f_UTRAN_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The UE is switched on.

    f_UT_SwitchOnUE (UT, false);

    //@siclog "Step 2-13" siclog@

    //Perform registration procedure

    f_UTRAN_GMMOnly_Attach (utran_Cell5);

    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC

    f_UTRAN_SendAuthCoOrdParameters(EUTRA);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get()));

    f_UTRAN_InterRAT_InitialiseAuthParams();

    //@siclog "Step 17" siclog@

    //Changes Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T2" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell7, -60, -62);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationCellOFF);

    t_WaitTime.start;

    //@siclog "Step 18-27" siclog@

    //Perform RAU procedure on cell 7

    f_UTRAN_RRC_ConnEst ( utran_Cell7 );

    f_UTRAN_GMM_RAU_MappedContext ( utran_Cell7 );

    f_UTRAN_RRC_ConnRel ( utran_Cell7, cell_Fach_Dcch );

    t_WaitTime.stop;

    //@siclog "Step 28" siclog@

    //Change Cell 1, Cell 5 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);

    f_UTRAN_SetCellPower(utran_Cell7, -70, -72);

    t_WaitTime.start;

    //@siclog "Step 29-38" siclog@

    //Perform RAU procedure on cell 5

    f_UTRAN_RRC_ConnEst ( utran_Cell5 );

    f_UTRAN_GMM_RAU_MappedContext ( utran_Cell5 );

    f_UTRAN_RRC_ConnRel ( utran_Cell5, cell_Fach_Dcch );

    t_WaitTime.stop;

    //@siclog "Step 39" siclog@

    //Change Cell 1, Cell 2 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell7, tsc_AttenuationCellOFF);

    //Tell EUTRA PTC to continue with test case by sending auth params

    f_UTRAN_SendAuthCoOrdParameters(EUTRA);

    f_UTRAN_TestBody_Set (false);

    //Wait for confirmation that test case has ended

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    //Release the UTRA cells

    f_UTRAN_ReleaseCell (utran_Cell5, f_UTRAN_CellInfo_GetConfigType (utran_Cell5) );

    f_UTRAN_ReleaseCell (utran_Cell7, f_UTRAN_CellInfo_GetConfigType (utran_Cell7) );

  }


After change :

	  function f_TC_6_3_3_UTRAN ( ) runs on UTRAN_PTC {

    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;

    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;

    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());

    timer t_WaitTime := tsc_IdleMode_GenericTimer_UTRAN;

    var SysInfoType1 v_SIB1_Cell7; //NOTE1
    //Initialise all cells

    f_UTRAN_Init (EUTRA_UTRAN);

    //Set RAC of cell 7 to different from RAC of cell 5 according to Pre-test condition: "Cell 5 and Cell 7 have different Routing Area."

    f_UTRAN_CellInfo_SetRAC(utran_Cell7, tsc_RAC_2);

    v_SIB1_Cell7 := f_UTRAN_SysInfo_GetSIB1 (utran_Cell7); //NOTE1
    v_SIB1_Cell7.cn_DomainSysInfoList[0].cn_Type.gsm_MAP := tsc_RAC_2 & bit2oct(tsc_NMO_I); //NOTE1
    f_UTRAN_SysInfo_SetSIB1 (utran_Cell7, v_SIB1_Cell7);   //NOTE1
    //Send the cell info to the EUTRA component

    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);

    //Get Cell Parameters for EUTRA cell

     v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);

    //Update SIB19 of cell 5 according to Table 6.2.3.3.3.3-1

    f_UTRAN_SysInfo_SetSIB19 (utran_Cell5,

                              cs_SIB19_UtraAndOneEutraEntry (cs_Utra_PriorityInfoList_Def (v_UtraPriorityValue),

                                                             cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                   v_EUTRA_MeasurementBandwidth)));

    f_UTRAN_SysInfo_SetSIB19 (utran_Cell7,

                              cs_SIB19_UtraAndOneEutraEntry (cs_Utra_PriorityInfoList_Def ( v_UtraPriorityValue ),

                                                             cs_Eutra_FreqAndPriList_OneEntry_Def (v_EUTRASysInfo.Eutra[0].Arfcn,

                                                                                                   v_EUTRA_MeasurementBandwidth)));

    //Create and configure Cell 5 & 7

    //Cell started switched on with default power level, -60 and max attenuation

    f_UTRAN_SS_CreateCellFACH (utran_Cell5);

    f_UTRAN_SS_CreateCellFACH (utran_Cell7);

    //Start sending System Information

    f_UTRAN_SendDefSysInfo (utran_Cell5);

    f_UTRAN_SendDefSysInfo (utran_Cell7);

    // Wait for CSG list to be emptied in LTE side

    f_IRAT_WaitForCoOrd_Trigger(EUTRA); //NOTE2
    //Set power levels according to T0

    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);

    f_UTRAN_TestBody_Set (true);

    //@siclog "Step 1" siclog@

    //The UE is switched on.

    f_UT_SwitchOnUE (UT, false);

    //@siclog "Step 2-13" siclog@

    //Perform registration procedure

    f_UTRAN_GMMOnly_Attach (utran_Cell5);

    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC

    f_UTRAN_SendAuthCoOrdParameters(EUTRA);

    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get()));

    f_UTRAN_InterRAT_InitialiseAuthParams();

    //@siclog "Step 17" siclog@

    //Changes Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T2" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell7, -60, -62);

    f_UTRAN_SetCellPower (utran_Cell5, tsc_AttenuationCellOFF);

    t_WaitTime.start;

    //@siclog "Step 18-27" siclog@

    //Perform RAU procedure on cell 7

    f_UTRAN_PrepareCellRRC_ConnEst (utran_Cell7, utran_Cell5); //NOTE3
    f_UTRAN_RRC_ConnEst ( utran_Cell7 );

    f_UTRAN_GMM_RAU_MappedContext ( utran_Cell7 );

    f_UTRAN_RRC_ConnRel ( utran_Cell7, cell_Fach_Dcch );

    t_WaitTime.stop;

    //@siclog "Step 28" siclog@

    //Change Cell 1, Cell 5 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);

    f_UTRAN_SetCellPower(utran_Cell7, -70, -72);

    t_WaitTime.start;

    //@siclog "Step 29-38" siclog@

    //Perform RAU procedure on cell 5

    f_UTRAN_PrepareCellRRC_ConnEst (utran_Cell5, utran_Cell7); //NOTE3
    f_UTRAN_RRC_ConnEst ( utran_Cell5 );

    f_UTRAN_GMM_RAU_MappedContext ( utran_Cell5 );

    f_UTRAN_RRC_ConnRel ( utran_Cell5, cell_Fach_Dcch );

    t_WaitTime.stop;

    //@siclog "Step 39" siclog@

    //Change Cell 1, Cell 2 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.

    f_UTRAN_SetCellPower(utran_Cell7, tsc_AttenuationCellOFF);

    //Tell EUTRA PTC to continue with test case by sending auth params

    f_UTRAN_SendAuthCoOrdParameters(EUTRA);

    f_UTRAN_TestBody_Set (false);

    //Wait for confirmation that test case has ended

    f_IRAT_WaitForCoOrd_Trigger (EUTRA);

    //Release the UTRA cells

    f_UTRAN_ReleaseCell (utran_Cell5, f_UTRAN_CellInfo_GetConfigType (utran_Cell5) );

    f_UTRAN_ReleaseCell (utran_Cell7, f_UTRAN_CellInfo_GetConfigType (utran_Cell7) );

  }


Execution Log Files

Nvidia Icera i500 UE

The Nvidia Icera i500 UE passed this test case on the Anritsu ME7832L LTE system with NAS and AS Snow 3G integrity and ciphering algorithm. The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file :

In the log file (in html format) the complete test case execution can be seen. All message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file
References

	[1]
	R5s130923: This archive comprises text format execution log file.

